
1. Why Use SAS?

This document provides the necessary background information needed to get started using SAS for your econometrics class.  The document should be used as a resource to help you learn basic data creation and analysis tools.

SAS may not be the easiest statistical package to learn how to use, but it has a number of very nice features: 

· Can handle very large data sets

· Is a very popular and powerful statistical package

· Employers like to see SAS on a resume

· You need it for your class

2. Getting Started

Starting SAS

To start SAS in Windows NT:

 Click on Start, Programs, The SAS System, The SAS System for Windows v6.12.

SAS Windows

Generally, you will use three main windows in SAS:

· Program editor:  Use this window to compose and edit your SAS programs.  Text in this window is generally saved with the .SAS extension.
· Log:  This is where your program log will be.  The program log will tell you whether your program ran successfully.  If there are any errors, they will appear in red. Text in this window is generally saved with the .LOG extension.
· Output: This is where any program output will be.  Text in this window is generally saved with the .LST extension.
Note:  You can toggle between the three windows by clicking on Window on the toolbar or by using the function keys:  F5 (Program editor), F6 (Log Window), F7 (Output window).

Note:  To save the text in any of the windows, click on File on the toolbar, then Save as…

Annoying Note:  When you change the “Save as type”, you will have to manually change the file extension.  SAS often defaults to the last file extension used.

Using Programs to Analyze Data in SAS

Unlike with some other statistical software packages, you will usually need to write programs in order to analyze your data.  The programs read in data, modify the data as needed, and perform statistical analysis.  Data is read in and modified in data steps, and data is analyzed using procedures.  

IMPORTANT: every line in a SAS program ends with a semicolon.

3. Data

You can think of SAS data in terms of spreadsheet data.  Columns are variables and each row is an observation.  Each observation contains a value for every variable associated with the data set.


A
B
C
D
E

1

Var 1
Var 2
Var 3
Var 4

2

Name
sex
age
dist

3
obs 1
Wendy
F
15
5

4
obs 2
Alex
M
17
15

5
obs 3
Amir
M
14
1

6
obs 4
Becky
F
17
4

7
obs 5
Alicia
F
16
30


In this example, each observation is a person.  There are four variables: name, sex, age, and distance to work.

Because SAS has its own way of storing data in binary files, imported ASCII data must be converted into SAS data before analysis can begin.  SAS allows you to access and use a number of data sets within the same program.  You will use two types of SAS data sets in your programs:

1) Permanent data sets are data sets that are stored in a user-designated directory (e.g. a:\mydata.sd2).  Permanent data sets are particularly useful if the data they contain will be used in a future application.  Permanent data sets will have the extension ‘.sd2’ on the PC.

2) Temporary data sets are data sets that will be automatically deleted after you finish your SAS session.  Temporary data sets are used when your only reference to that data will be within the current program.

The most recently created data set is called the active data set.  This data set includes the initial variables and values and any newly created variables or modified values that will be processed by the SAS program.  The active data set may be permanent or temporary.

Importing ASCII data

To import ASCII (text) data, you first need an ASCII data set.  Typically, for the econometrics class, you will have to download the ASCII data from the course web page.

Accessing raw data from a web page

1) Navigate to the appropriate web page.

2) Select the data.  If the first row contains variable names, do not select this row.

3) Select File, Save As… from the toolbar.  For the Save as type, choose “All Files” (Netscape Navigator) or “Text File” (Internet Explorer).  

4) Enter a name and save your data (typically with a “.txt” extension).

Note: In some cases, the data on a web page will be stored as a compressed (zipped) file.  If that is the case, all you will need to do is select a location to save the file to when prompted.  If the file has a ‘.zip’ extension, you will need to decompress the file using PKZIP or WINZIP.  If the file has an ‘.exe’ extension, merely double click on the file name to uncompress.

Once an ASCII data set exists, you are ready to write a program that imports and uses the data.  The first step is to create filename, which tells SAS where your ASCII data set is.

filename statement creates a logical name (an alias or nickname) for an ASCII data file.

filename statetxt 'c:\users\states.txt';

If you intend to use or create a permanent data set, you will also need to create a libname, which lets you reference a data directory.

libname statement establishes a logical name (an alias or nickname) for a directory which contains (or will contain) a permanent SAS data set.  

libname mydisk ‘a:\’;

libname users ‘c:\users\’;

Note:  A program may contain a number of libnames and filenames.

Note:  If one of your libnames refers to the floppy disk drive (like “mydisk”), you must have a disk in the drive.

Data Step

Data steps are the place where we create new data files, add new data, delete data, or transform data.  Data steps always begin with the word “data”.  After the word data comes the name of the data set you are creating.  Temporary data sets are named with only one word.

data states;

Tempdat is a temporary data set that will only exist during our current session.  If we want to create a permanent data set, we must include a reference to the logical name in the data set’s name.  We do this by including the logical name followed by a period followed by the name of the data set.

data users.stperm;

In this example, we will create c:\users\stperm.sd2 with the data statement since “users” refers to ‘c:\users’.

Note:  The name of your data set must start with a character and must be no longer than eight characters.

infile statement opens an ASCII data set 

input statement describes the arrangement of values in a data file and assigns the values to SAS variables.  It is here that you assign the variable names, so the first row of your data should NOT contain the variable names.

infile statetxt;

  input state $ 1-2 pop 12-19 inc_cp 20-25 spend 28-32;
The infile command opens c:\users\states.txt.

The input command reads the character variable state from columns 1-2, the numeric variables pop, inc_cp, and spend from columns 12-19, 20-25, and 28-32 from the data set c:\users\states.txt.  Variable names followed by a ‘$’ are character variables and those without a ‘$’ are numeric.

Note:  One way to check where the columns for a variable start is to bring the data into Word.  Word displays the column number at the bottom of the screen.

Note:  The names of your variables set must start with a character and must be no longer than eight characters

Bringing in SAS data

Bringing in SAS data is much easier than importing ASCII data.  If the data already exists as a temporary or permanent SAS data set, we do not need to use the infile or input statements.  Instead we use the set statement.

set statement reads a permanent SAS data set into a new SAS data set.  Set can refer to either a temporary or a permanent SAS data set.

data states2;

  set states;

data getmore;

  set users.stperm;

Data manipulation

Data can be manipulated only within a SAS data step.  Inside the data step, the values of variables may be modified or new variables may be created.  

spend_cp = spend/pop;

The above line creates a per capita spending variable.

if-then

Data manipulation usually involves making a decision or calculation based on the value of a variable with respect to a particular value or the value of another variable.  Typically, this is done with the ‘if-then’ statement and its subcommands ‘and’, ‘or’, and ‘else’.  There are several other operands associated with the ‘if-then’ statement:

eq 
or
=
Equals

ne
or
~=
Not equal

gt
or
>
Greater than

ge
or
>=
Greater than or equal

lt
or
<
Less than

le
or
<=
Less than or equal

Note:  Numeric and character variables may not be compared to one another with ‘if-then’ statements.  When character values are used, they must be enclosed in single or double quotes.  

Creating new variables

if-then statements are often used to create a new variable:

if var_a = <value> then var_b = <some value>;

This is an ideal way to create a dummy variable, which usually takes on a value of 1 when a condition is true, and 0 when the condition is false.

bigpop = 0;

if pop > 5000 then bigpop =1;

The above statements create the dummy variable bigpop.  bigpop = 1 only when population is greater than 5,000,000 (population is measured in thousands).

Changing the value of the same variable

if-then statements can also be used to change the value of the same variable:

if variable = <value> then variable = <new value>;

For example, we might want to set all negative values of a spending measure to zero:

if spend < 0 then spend = 0;

Linking conditions

The ‘and’ and ‘or’ subcommands can be used to link several conditions in an ‘if-then’ statement.  All of the conditions must be true for the command to execute.

if (state = ‘NY’ or state = ‘NJ’) and pop > 5000 then newbpop=1;

This command sets newbpop equal to 1 if an observation is from New York or New Jersey and population is greater than 5,000,000.

Else subcommand

The else subcommand issues instructions if the condition(s) in the ‘if-then’ statement is/are false.

if (state = ‘NY’ or state = ‘NJ’) and pop > 5000 then newbpop=1;

  else newbpop=0;
Here, the else subcommand sets newbpop equal to 0 for all observations that are not in New York or New Jersey  or for observations that are in New York or New Jersey and have populations less than 5,000,000.

Keeping and dropping variables

Within a data statement, it is possible to drop variables you no longer need or to keep just the variables you want.

keep var1 var2 var3;  (no commas between the variable names)

The above keep statement will keep only variables var1, var2, and var3 in the active data set – all other variables are dropped.

drop var2 var3; 

The above drop statement will drop variables var2 and var3.  All other variables in the active data set will be kept.

Note:  You can also use the keep and drop statements in the data or set statements:

data states3 (drop = pop);

or

set states (keep = state spend pop);

Keeping and dropping observations

Just as you can delete variables, it is also possible to delete observations.

if condition then delete;

if spend >25000 then delete;

The above statement deletes all observations for which spending is greater than $25,000,000,000  (spending figures are in millions)

FYI:

Using if with set

Another way to keep just some observations is to use an if statement with a set statement:

data smlspend;

   set states;

     if spend <=4000;

The above data statements creates a temporary data set called smlspend  that contains data from states only if spending is less than or equal to 4,000,000,000.

4. Procedures

Procedures are the reason why you will want to use SAS.  All statistical operations take place in the procedures or ‘PROC’ step of a SAS program. Available procedures range from descriptive statistics (e.g. frequency distributions) to inferential statistics (e.g. regressions).  The names of these procedures are generally mnemonic.  The procedures themselves contain many options that are useful in tailoring your output.  See the manual “SAS Procedures” for a complete list of SAS procedures and their options.

Note:  Each of these examples ends with the statement ‘run;’.  You need to add the ‘run’ statement if you are running SAS interactively.  If you are running the commands in a program, you must have a ‘run’ statement at the end of the program.

Note: To execute/submit a program or a command from the program window, click on the button with the little person running, or hit F8.

CONTENTS

Proc contents provides a summary of the data set, including the list of variables and number of observations.

proc contents data = states;

run;

FREQ

Proc freq provides a frequency distribution of the variables specified in the tables subcommand.

proc freq data=states;

  tables pop spend;

  tables pop*spend;

run;

The first command informs SAS that a frequency distribution is requested.

The subcommand ‘tables pop spend’ instructs SAS to produce a frequency distribution for both variables pop and spend.

The second subcommand ‘tables pop*spend’ instructs SAS to produce a crosstabulation for the variables pop and spend.

UNIVARIATE

Proc univariate provides descriptive statistics for the variables you specify.  The statistics provided are mean, median, mode, standard deviation, quantiles, minimum and maximum values, and the five largest and smallest observations in the data set.

Proc univariate data=states;

  var pop;

run;

Note:  The var subcommand for this procedure and most other procedures tells SAS on which variable to perform the procedure.  If I had omitted the subcommand, SAS would have performed the univariate procedure on all variables in the data set.

PRINT

Proc print prints out the data to the screen.

Proc print data=states;

  var state inc_cp spend;

run;
MEANS

Proc means provides (by default) the number of observations, mean, standard deviation, minimum and maximum values.

Proc means data=states;

  var inc_cp spend;

run;
Note: to specify which statistics you would like reported, place the name of that statistic or statistics right after the word ‘means’.  For example, to only report the mean:

Proc means mean data=states;

  var inc_cp spend;

run;
CORR

Proc corr provides a correlation matrix for the variables specifies.

proc corr data=states;

 var spend pop;

run;
Note:  To include a covariance matrix in output, include the option cov:

proc corr cov data=states;

 var spend pop;

run;
REG

Proc reg estimates regressions.

Proc reg data=states;

 model spend = inc_cp pop;

run;

The dependent variable goes right after the word ‘model’ and the independent variables follow the equals sign.

SORT

Proc sort sorts the data by a particular variable.

Proc sort data=states;

  by state;

run;

Note:  by default, SAS sorts in ascending order (smallest to largest).  For character variables, such as state, SAS will sort in ascending alphabetical order (a to z).  If you need to sort in descending order, place the word DESCENDING before the variable name:

Proc sort data=states;

  by DESCENDING state;

run;
Note:  SAS can also sort by more than one variable:

Proc sort data=states;

  by pop spend;

run;
In this example, SAS sorts first on population, then on spending.

FSBROWSE

Proc fsbrowse allows you to look at the data one observation at a time in a special window.  Use the ‘Page Up’ and ‘Page Down’ keys to move from observation to observation.

Proc fsbrowse data= states;

run;

Tip:  To look at a SAS data set in a spreadsheet format, double click on the SAS file name. This will pull the data up into a special SAS window.

Tip:  If you double click on a SAS program file, SAS will launch and run the program.

By

Almost all procedures can include a by statement.  This splits up the procedure by the ‘by’ variable.  Before you use the by statement, the data must first be sorted by that variable(s).

proc sort data = states;

  by state;

proc means data=states;

  by state;

run;

5. Where to get help

SAS manuals

There are a number of SAS manuals that are helpful.  In particular, seek out the Procedures Guide and SAS Language.  The CCon has a book that you can check out that that combines the two:  SAS Language and Procedures.  If you get stuck or have questions, this is a good place to start.

SAS help on line

SAS has on line help that includes many sample programs and SAS/TUTOR online training

Ask an expert

If you can’t get unstuck, find someone who knows SAS.  Often, if you show them the log file, they’ll be able to quickly identify your problem.

6. Other

Editing your programs and output

All SAS files (other than SAS data sets) are ASCII data.  Therefore, if you prefer, you can write your programs in your favorite word processor and save the file as text with a ‘.sas’ extension.  (If you use Word, use double quotes around the name of your program name to prevent Word from adding a ‘.txt’ extension: “myprog.sas”.)

Likewise, you will most likely want to bring your SAS output into a word processor in order to edit it.  You can either cut and paste from the output window, or you can open the saved .lst file in the word processor.  Also note that one of Word’s fonts is SAS MONOSPACE FONT – you can use this font to make the SAS output much more legible.

Writing good code

SAS allows a great deal of freedom with regard to how you write your SAS code.  Here are some stylistic guidelines that are good to follow:

· Use indentation to show logical structure.

· Use space and skipped lines to improve readability.

· Write only one SAS statement on a line.

· Use capitalization to set certain key words apart.

· Show data set names explicitly, even when SAS will default to the last data set created.  This can avoid a lot a future confusion.

· Use comments (/* this comments out  */).

Common errors 

· Forgetting to use a semicolon at the end of the statement

· Confusing character (string) and numeric variables

· Not sorting your data before using BY before merging

· Forgetting to end with a ‘run’ statement

· Forgetting to close the program document in Word before running the program

Some Useful Functions

FUNCTION
RETURNS
ABS

Absolute value

EXP

Exponential

LOG

Natural logarithm

LOG10
Logarithm to base 10

MAX

Largest value

MIN

Smallest value

MOD

Remainder value

SQRT

Square Root

Types of Files

Extension
File Type

Data

.txt
An ASCII data file

.sd2
SAS data

Other

.sas
SAS program

.log
SAS log

.lst
SAS output

Function Keys

I used F9 to generate this list of function keys.  F4 is the function key I use the most.  I use it to recall the last program that I submitted.

F1       help


Starts the help facility

F2       reshow

F3       end; /*gsubmit buffer=default*/

F4       recall

Recalls last commands to the program editor

F5       pgm


Takes you to the ‘program editor’ window

F6       log


Takes you to the ‘log’ window

F7       output

Takes you to the ‘output’ window

F8       zoom off;submit Submits commands from the program editor

F9       keys


Displays function keys

F11      command bar

F12

Acquiring SAS

For information regarding getting your own (possibly free) copy of SAS, visit the Heinz Computing software web page:

http://www.heinz.cmu.edu/heinz/computing/studentsoftware.html

Sample program

/* This program USESTATES.SAS uses the income and spending data from the 48 states */

/* in the STATES.SD2 dataset we created last week */

/* The program also performs some basic statistical analysis */

options ls=80 ps=400;

/* this optional line sets the line size (how many characters per line) and the

    page size (how many lines per page) */

libname rs 'r:\academic\90770\review sessions';

/*this sets up a directory and names it rs */

data statetmp;

  set rs.states;  /* this calls up the states SAS data set */

    if pop > 5000 then bigpop = 1;

      else bigpop = 0;

/* let's find out what's in our data set */

proc contents;

/* let's view some descriptive statistics of our data */

proc means;

   var pop income gvt inccap gvtcap bigpop;

/* note, without the VAR statement, SAS would have given us stats for all of the variables */

/* VAR specifies which variables we want to see statistics for */

/* let's take a look at correlation and covariance between population and

per capita government spending */

proc corr cov;

 var pop bigpop gvtcap;

/* let's see if the mean governemnt spending per capita is different in big states

than in small states */

/* first, we need to sort by bigpop*/

proc sort;

  by bigpop;

proc means;

  var gvtcap;

  by bigpop;

/* Including 'by' in the means procedure gives us means for big and small states */

/* we have learned how to test this difference */

/* can we see a relationship between population and spending? */

/* finally, let's estimate regressions */

proc reg;

  model gvtcap = pop;

  model gvtcap = bigpop;

run;

7. Thanks 

Some of this material comes from Hugh Shinn’s PC SAS Class, Steven Appold’s SAS cheat sheets, and from CMU’s “Introduction (Intermediate, and Advanced) to SAS in a UNIX Environment” 

